
Reasoning (LPNMR-7), I. Niemelä and V. Lifschitz, editors, LNCS,c© 2004 Springer.

Towards Automated Integration of Guess and Check
Programs in Answer Set Programming?

Thomas Eiter1 and Axel Polleres2

1 Institut für Informationssysteme, TU Wien, A-1040 Wien, Austria
eiter@kr.tuwien.ac.at

Abstract. Many NP-complete problems can be encoded in the answer set se-
mantics of logic programs in a very concise way, where the encoding reflects
the typical “guess and check” nature ofNP problems: The property is encoded
in a way such that polynomial size certificates for it correspond to stable mod-
els of a program. However, the problem-solving capacity of full disjunctive logic
programs (DLPs) is beyondNP at the second level of the polynomial hierar-
chy. While problems there also have a “guess and check” structure, an encod-
ing in a DLP is often non-obvious, in particular if the “check” itself is co-NP-
complete; usually, such problems are solved by interleaving separate “guess” and
“check” programs, where the check is expressed by inconsistency of the check
program. We present general transformations of head-cycle free (extended) logic
programs into stratified disjunctive logic programs which enable one to integrate
such “guess” and “check” programs automatically into a single disjunctive logic
program. Our results complement recent results on meta-interpretation in ASP,
and extend methods and techniques for a declarative “guess and check” problem
solving paradigm through ASP.

1 Introduction

Answer set programming (ASP) [19, 7] is widely proposed as a useful tool for express-
ing properties inNP, where solutions and polynomial time proofs for such properties
correspond to answer sets of normal logic programs, which cover by well-known com-
plexity results the classNP. An example for such a property is whether a given graph
has a legal 3-coloring, where any such coloring is itself a certificate for this property.

However, we also might encounter situations in which we want to express a problem
which is complementary to someNP problem, and thus belongs to the class co-NP; it is
widely believed that in general, not all such problems are inNP and hence not always
a polynomial-size certificate checkable in polynomial time exists. One such problem
is, e.g., the property that a graph isnot 3-colorable. Such propertiesp can analogously
be expressed by a normal logic program (equivalently, by a head-cycle free disjunctive
logic program [1])Πp, where the property holds iffΠp has no answer set at all.

Checks in co-NP typically occur as subproblems within more complex problems
which have complexity higher thanNP, for instance:

Quantified Boolean Formulas (QBFs):Evaluating a QBF, where we have to check,
given a QBF of the form∃X∀Y Φ(X, Y ), and an assignmentσ to the variablesX,

whether∀Y Φ(σ(X), Y ) evaluates to true.

Strategic Companies:Checking whether a set of companies is strategic (cf. [11]).

Conformant Planning: Checking whether a given plan is conformant [8], provided
executability of actions is polynomially decidable (cf. [4, 22]).

Further examples can be found in [6, 5]. In general, the corresponding logic program
Πp for this check can be easily formulated and the overall problem (evaluating the QBF,
finding a strategic companies set resp. a conformant plan) solved in a 2-step approach:

1. Generate a candidate solution by means of a logic programΠguess.
2. Check the solution by another logic programΠcheck (=Πp).

However, it is often not clear how to combineΠguess andΠcheck into asinglepro-
gramΠsolve which solves the overall problem. Simply taking the unionΠguess∪Πcheck

does not work, and rewriting is needed. Theoretical results [6] informally give strong ev-
idence that for problems withΣP

2 -complexity, it is required thatΠcheck (given as a nor-
mal logic program or a head-cycle free disjunctive logic program) is rewritten into a dis-
junctive logic programΠ ′

check such that the answer sets ofΠsolve = Πguess ∪Π ′
check

yield the solutions of the problem, whereΠ ′
check emulates the inconsistency check for

Π ′
check as a minimal model check, which is co-NP-complete for disjunctive programs.

This becomes even more complicated by the fact thatΠ ′
check must not crucially rely

on the use of negation, since it is essentially determined by theΠguess part. These
difficulties can make rewritingΠcheck to Π ′

check a formidable and challenging task.
In this paper, we present a generic method for rewritingΠcheck automatically by

using a meta-interpreter approach. In particular, we make the following contributions:

(1) We provide a polynomial-time transformationtr(Π) from propositional head-
cycle-free [1] (extended) disjunctive logic programs (HDLPs)Π to disjunctive logic
programs (DLPs), such that the following conditions hold:

T1 Each answer setS′ of tr(Π) corresponds to an answer setS of Π, such that
S = {l | inS(l) ∈ S′} for some predicateinS(·).

T2 If the original program has no answer sets, thentr(Π) has exactly one designated
answer setΩ, which is easily recognizable.

T3 The transformation is of the formtr(Π) = F (Π) ∪ Πmeta, whereF (Π) is a
factual representation ofΠ andΠmeta is a fixedmeta-interpreter.

T4 tr(Π) is modular(at the syntactic level), i.e.,tr(Π) =
⋃

r∈Π tr(r). Moreover, it
is a stratified DLP [20, 21] and uses negation only in its “deterministic” part.

We also describe optimizations and a transformation to positive DLPs, and show that in
a precise sense, modular transformations to such programs do not exist.
(2) We show how to usetr(·) for integrating separate guess and check programs

Πguess andΠcheck, respectively, into a single DLPΠsolve such that the answer sets of
Πsolve yield the solutions of the overall problem.
(3) We demonstrate the method on the examples of QBFs and conformant plan-

ning [8] under fixed polynomial plan length (cf. [4, 22]), where our method proves to
loosen some restrictions of previous encodings.

Our work enlarges the range of techniques for expressing problems using ASP,
in a direction which to our knowledge has not been explored so far. It also comple-
ments recent results about meta-interpretation in ASP [16, 2, 3]. We fruitfully exploit

the construction oftr(·) to further elucidate the natural guess and check programming
paradigm for ASP, as discussed in [11] or in [14] (named “Generate/Define/Test” there),
and we fill a gap by providing an automated construction for integrating guess and check
programs. It is worth noticing that such an integration is non-trivial even for manual
construction in general. Apart from being pure ASP solutions, integrated encodings
may be straight subject to automated program optimization within ASP solvers, con-
sidering both the guess and check part as well as their interaction; this is not immediate
for separate programs.

For space constraints, most proofs and longer encodings are omitted here. All proofs
and further details (encodings, etc) are given in an extended version of this paper.3

2 Preliminaries
We assume that the reader is familiar with logic programming and answer set semantics
(see [7, 19]) and only briefly recall the necessary concepts.

A literal is an atoma(t1, . . . , tn),or its negation¬a(t1, . . . , tn), where “¬” (alias,
“–”) is the strong negation symbol, in a function-free first-order language with at least
one constant, which is customarily given by the programs considered. By|a| = |¬a| =
a we denote the atom of a literal. Extended disjunctive logic programs (EDLPs; or
simply programs) are finite setsΠ of rulesr

h1 v . . . v hl :− b1, . . . , bm, not bm+1, . . . not bn. (1)

l,m, n ≥ 0, where eachhi andbj is a literal andnot is weak negation (negation as
failure). ByH(r) = {h1, . . . , hl}, B+(r) = {b1, . . . , bm}, B−(r) = {bm+1, . . . , bn},
andB(r) = B+(r) ∪ B−(r) we denote the head and (pos., resp. neg.) body of ruler.
Rules with|H(r)|=1 andB(r)=∅ arefactsand rules withH(r)=∅ constraints. A rule
r is positive, if “ not” does not occur in it, andnormal, if |H(r)| ≤ 1. A programΠ is
positive(resp.normal) if all its rules are positive (resp., normal). We omit “extended”
in what follows and refer to EDLPs as DLPs etc.

Literals (resp. rules, programs) areground if they are variable-free. Non-ground
rules (resp. programs) amount to theirground instantiation, i.e., all rules obtained by
substituting variables with constants from the (implicit) language.

A ground programΠ is head-cycle free[1], if no literals l 6= l′ occurring in the
same rule head mutually depend on each other by positive recursion;Π is stratified [20,
21], if no literal l depends by recursion through negation on itself.

Recall that theanswer set semantics[7] for DLPs is as follows. Denote byLit(Π)
the set of all ground literals for a programΠ. Then,S is ananswer setof Π, if S is a
minimal (under⊆) consistent4 setS ⊆ Lit(Π) satisfying all rules in the reductΠS ,
which contains all rulesh1 v . . . v hl :− b1, . . . , bm for all ground instances of rules
(1) in Π such thatS ∩B−(r) = ∅.

3 Meta-Interpreter Transformation
As mentioned above, a rewriting of a given programΠcheck to a programΠ ′

check for
integrating a guess and a check part into a single program is tricky in general. The

3 See http://www.kr.tuwien.ac.at/staff/axel/guessncheck/ (forthcoming).
4 For our concerns, we disregard a possible inconsistent answer set.

working of the answer set semantics is not easy to be emulated inΠ ′
check, since essen-

tially we lack negation inΠ ′
check: Upon a “guess”S for an answer set ofΠsolve =

Πguess ∪ Π ′
check, the reductΠS

solve is not-free. Thus, contrary toΠcheck, there is no
possibility to consider varying guesses for the value of negated atoms inΠ ′

check in
combination with one guess for the negated atoms inΠguess in the combined program
Πsolve. On the other hand, if there is no disjunction inΠ ′

check thenΠsolve is Horn;
thus, its answer sets can be guessed and checked inNP.

This leads us to consider an approach in which the programΠ ′
check is constructed

by the use of meta-interpretation techniques [16, 2, 3]: the idea is that a programΠ
is represented by a set of facts,F (Π), which is input to a fixed programΠmeta, the
meta-interpreter, such that the answer sets ofΠmeta ∪ F (Π) correspond to the answer
sets ofΠ. Note that existing meta-interpreters are normal logic programs, and can not
be used for our purposes for the reasons explained above; we have to construct a novel
meta-interpreter which is essentiallynot-free and contains disjunction. To this end, we
exploit the following characterization of (consistent) answer sets for HDLPs:

Theorem 1 (cf. [1]).For any ground HDLPΠ, a consistentS ⊆ Lit(Π) is an answer
set ofΠ iff (1) S satisfiesΠ and (2) there is a functionφ : Lit(Π) → {0, 1, 2, . . .} such
that for each literall∈S, there is a ruler∈Π with (a) B+(r) ⊆ S , (b) B−(r)∩S = ∅,
(c) l ∈ H(r), (d) S ∩ (H(r) \ {l}) = ∅, and (e) φ(l′) < φ(l) for eachl′∈B+(r).

Theorem 1 will now serve as a basis for a transformation from a given HDLPΠ to a
DLP tr(Π) = F (Π) ∪Πmeta such thattr(Π) fulfills the propertiesT1–T4:

Input representation F (Π) As input for the meta-interpreterΠmeta below, we choose
the following representationF (Π) of the propositional programΠ.

We assume that each ruler has a unique namen(r) as usual; for convenience, we
identify r with n(r). For any ruler ∈ Π, we set up inF (Π) the following facts:

lit(h, l, r). atom(l, |l|). for each literall ∈ H(r),
lit(p, l, r). for each literall ∈ B+(r),
lit(n, l,r). for each literall ∈ B−(r).

While the facts for predicatelit obviously encode the rules ofΠ, the facts for predicate
atom indicate whether a literal is classically positive or negative. We only need this
information for head literals; this will be further explained below.

Meta-Interpreter Πmeta We construct our meta-interpreter programΠmeta, which
in essence is a positive disjunctive program, in a sequence of several steps. They center
around checking whether a guess for an answer setS ⊆ Lit(Π), encoded by a predicate
inS(·), is an answer set ofΠ by testing the criteria of Theorem 1. The steps of the
transformation cast the conditions of the theorem into rules ofΠmeta, and provide
auxiliary machinery for this aim.

Step 1We add the following preprocessing rules:
1: rule(L, R) :− lit(h, L, R), not lit(p, L, R), not lit(n, L, R).
2: ruleBefore(L, R) :− rule(L, R), rule(L, R1), R1 < R.
3: ruleAfter(L, R) :− rule(L, R), rule(L, R1), R < R1.
4: ruleBetween(L, R1, R2) :− rule(L, R1), rule(L, R2), rule(L, R3),

R1 < R3, R3 < R2.

5: firstRule(L, R) :− rule(L, R), not ruleBefore(L, R).
6: lastRule(L, R) :− rule(L, R), not ruleAfter(L, R).
7: nextRule(L, R1, R2) :− rule(L, R1), rule(L, R2), R1 < R2,

not ruleBetween(L, R1, R2).

8: before(HPN, L, R) :− lit(HPN, L, R), lit(HPN, L1, R), L1 < L.
9: after(HPN, L, R) :− lit(HPN, L, R), lit(HPN, L1, R), L < L1.

10: between(HPN, L, L2, R) :− lit(HPN, L, R), lit(HPN, L1, R),
lit(HPN, L2, R), L < L1, L1 < L2.

11: next(HPN, L, L1, R) :− lit(HPN, L, R), lit(HPN, L1, R), L < L1,
not between(HPN, L, L1, R).

12: first(HPN, L, R) :− lit(HPN, L, R), not before(HPN, L, R).
13: last(HPN, L, R) :− lit(HPN, L, R), not after(HPN, L, R).

14: hlit(L) :− rule(L, R).

Lines 1 to 7 fix an enumeration of the rules inΠ from which a literall may be derived,
assuming a given order< on rule names (e.g. inDLV [11], built-in lexicographic order;
< can also be easily generated using guessing rules). Note that under answer set seman-
tics, we need only to consider rules where the literall to prove does not occur in the
body. Next, lines 8 to 13 fix enumerations ofH(r), B+(r) andB−(r) for each rule.
The final line 14 collects all literals that can be derived from rule heads. Note that lines
1-14 plusF (Π) form a stratified program, which has a single answer set (cf. [20, 21]).

Step 2 We add rules which “guess” a candidate answer setS ⊆ Lit(Π) and a total
orderingphi onS corresponding with the functionφ in Theorem 1.(2).

15: inS(L) v ninS(L):−hlit(L).
16: ninS(L):−lit(pn, L, R), not hlit(L). for eachpn ∈ {p, n}
17: notok:−inS(L), inS(NL), L !=NL, atom(L, A), atom(NL, A).
18: phi(L, L1) v phi(L1, L):−inS(L), inS(L1), L < L1.
19: phi(L, L2):−phi(L, L1), phi(L1, L2).

Line 15 focuses the guess ofS to literals occurring in some relevant rule head inΠ;
other literals can not belong toS (line 16). Line 17 then checks whetherS is consistent,
deriving a new distinct atomnotok otherwise. Line 18 guesses a strict total order
phi on inS where line 19 guarantees transitivity; note that minimality of answer sets
prevents thatphi is cyclic, i.e., thatphi(L, L) holds.

In the subsequent steps, we check whetherS and phi violate the conditions of
Theorem 1 by deriving the distinct atomnotok in case, indicating thatS is not an
answer set orphi does not represent a proper functionφ.

Step 3Corresponding to condition1 in Theorem 1,notok is derived whenever there is
an unsatisfied rule by the following program part:

20: allInSUpto(p, Min, R):−inS(Min), first(p, Min, R).
21: allInSUpto(p, L1, R):− inS(L1), allInSUpto(p, L, R), next(p, L, L1, R).
22: allInS(p, R):−allInSUpto(p, Max, R), last(p, Max, R).
23: allNinSUpto(hn, Min, R):−ninS(Min), first(hn, Min, R).
24: allNinSUpto(hn, L1, R):− ninS(L1), allNinSUpto(hn, L, R),

next(hn, L, L1, R).
25: allNinS(hn, R):−allNinSUpto(hn, Max, R), last(hn, Max, R).


for each
hn∈{h, n}

26: hasHead(R):−lit(h, L, R).
27: hasPBody(R):−lit(p, L, R).
28: hasNBody(R):−lit(n, L, R).
29: allNinS(h, R):−lit(HPN, L, R), not hasHead(R).
30: allInS(p, R):−lit(HPN, L, R), not hasPBody(R).
31: allNinS(n, R):−lit(HPN, L, R), not hasNBody(R).
32: notok:−allNinS(h, R), allInS(p, R), allNinS(n, R), lit(HPN, L, R).
These rules compute by iteration overB+(r) (resp.H(r), B−(r)) for each ruler,

whether for all positive body (resp. head and weakly negated body) literals in ruler
inS holds (resp.ninS holds) (lines 20 to 25). Here, empty heads (resp. bodies) are
interpreted as unsatisfied (resp. satisfied), cf. lines 26 to 31. The final rule 32 fires
exactly if one of the original rules fromΠ is unsatisfied.

Step 4We derivenotok whenever there is a literall ∈ S which is not provable by any
rule r wrt. phi. This corresponds to checking condition2 from Theorem 1.

33: failsToProve(L, R):−rule(L, R), lit(p, L1, R), ninS(L1).
34: failsToProve(L, R):−rule(L, R), lit(n, L1, R), inS(L1).
35: failsToProve(L, R):−rule(L, R), rule(L1, R), inS(L1), L1 !=L, inS(L).
36: failsToProve(L, R):−rule(L, R), lit(p, L1, R), phi(L1, L).
37: allFailUpto(L, R):−failsToProve(L, R), firstRule(L, R).
38: allFailUpto(L, R1):− failsToProve(L, R1), allFailUpto(L, R),

nextRule(L, R, R1).
39: notok:−allFailUpto(L, R), lastRule(L, R), inS(L).

Lines 33 and 34 check whether condition2.(a) or (b) are violated, i.e. some rule
can only prove a literal if its body is satisfied. Condition2.(d) is checked in line 35, i.e.
r fails to provel if there is somel′ 6= l such thatl′ ∈ H(r)∩S. Violations of condition
2.(e) are checked in line 36. Finally, lines 37 to 39 derivenotok if all rules fail to prove
some literall ∈ S by iterating over all rules withl ∈ H(r) using the order from Step 1.
Thus, condition2.(c) is implicitly checked.

Step 5Whenevernotok is derived, indicating a wrong guess, then we apply a saturation
technique as in [6, 12] to some other predicates, such that a canonical setΩ results. This
set turns out to be an answer set iff no guess forS andφ works out, i.e.,Π has no answer
set. In particular, we saturate the predicatesinS, ninS, andphi by the following rules:

40: phi(L, L1):−notok, hlit(L), hlit(L1).
41: inS(L):−notok, hlit(L).
42: ninS(L):−notok, hlit(L).

Intuitively, by these rules, any answer set containingnotok is “blown up” to an
answer setΩ containing all possible guesses forinS, ninS, andphi.

3.1 Answer Set Correspondence

Let tr(Π) = F (Π)∪Πmeta, whereF (Π) andΠmeta are the input representation and
meta-interpreter as defined above. Clearly,tr(Π) satisfies propertyT3, and as easily
checked,tr(Π) is modular. Moreover,¬ does not occur intr(Π) andnot only strati-
fied. The latter is not applied to literals depending on disjunction; it thus occurs only in
the deterministic part oftr(Π), i.e.T4 holds.

To establishT1 andT2, we define the literal setΩ as follows:

Definition 1. LetΠi
meta be the set of rules inΠmeta established in Stepi ∈ {1, . . . , 5}.

For any programΠ, let ΠΩ = F (Π) ∪
⋃

i∈{1,3,4,5} Πi
meta ∪ {notok.}. Then,Ω is

defined as the answer set ofΠΩ .

The fact thatΠΩ is a stratified normal logic program without¬ and constraints, which
as well-known has a single answer set, yields the following lemma.

Lemma 1. Ω is well-defined and uniquely determined byΠ.

Theorem 2. For any given HDLPΠ the following holds fortr(Π):
1. tr(Π) has some answer set, andS′ ⊆ Ω for any answer setS′ of tr(Π).
2. S is an answer set ofΠ if and only if there exists an answer setS′ of tr(Π) such

thatS = {l | inS(l) ∈ S′} andnotok 6∈ S′.
3. Π has no answer set if and only iftr(Π) has the unique answer setΩ.

The following proposition is not difficult to establish.

Proposition 1. GivenΠ, the transformationtr(Π), as well as the ground instantiation
of tr(Π), is computable inLOGSPACE(thus in polynomial time).

Note thattr(Π) is not polynomial faithful modular (PFM) in the sense of [9]: (i)
T1 does not claim a strict one-to-one correspondence between the answer sets ofΠ
andtr(Π). Indeed,tr(Π) might have several answer sets corresponding to a particular
answer setS of Π, reflecting different possible guesses forφ. (ii) Faithfulness as in [9]
conflicts with propertyT2.

As noticed above,tr(Π) uses weak negation only stratified and in a deterministic
part of the program; we can easily eliminate it by computing in the transformation the
complement of each predicate accessed throughnot and providing it inF (Π) as facts;
we then obtain a positive program. (The built-in predicates< and != can be eliminated
similarly if desired.) However, this modified transformation is not modular. As shown
next, this is not incidental.

Proposition 2. There is no modular transformationtr′(Π) from HDLPs to DLPs sat-
isfyingT1, T2 andT3 such thattr′(Π) is a positive program.

Proof. Assuming that such a transformationtr′(Π) exists, we derive a contradiction.
Let Π1 = {a :− not b.} andΠ2 = Π1 ∪ {b.}. Then,tr′(Π2) has some answer set
S2. Sincetr′(·) is modular,tr′(Π1) ⊆ tr′(Π2) holds and thusS2 satisfies each rule
in tr′(Π1). Hence,S2 contains some answer setS1. By T1, inS(a) ∈ S1 must hold,
and henceinS(a) ∈ S2. By T1 again, it follows thatΠ2 has an answer setS such that
a ∈ S. But the single answer set ofΠ2 is {b}, a contradiction. ut

We remark that Proposition 2 remains true ifT1 is generalized such that the answer
setS of Π corresponding toS′ is given byS = {l | S′ |= Φ(l)}, whereΦ(x) is
a monotone query (e.g., computed by a normal positive program without constraints).
Moreover, if a successor predicatenext(X, Y) and predicatesfirst(X) andlast(X)
for the constants are available (on a finite universe, resp. the constants inΠ and rule
names), then the negation of the non-input predicates accessed throughnot can be
computed by a positive normal program, since such programs capture polynomial time
computability by well-known results on the expressive power of Datalog [18]; thus,
negation of input predicates inF (Π) is sufficient in this case.

3.2 Optimizations
Πmeta can be modified in several respects. We discuss here some modifications which,
though not necessarily shrinking size of the transformation, intuitively prune the search
of an answer set solver applied totr(Π). The extended paper considers further ones.

(OPT1)Give up modularityIf we sacrifice modularity (i.e. thattr(Π) =
⋃

r∈Π tr(r)),
and allow thatΠmeta partly depends on the input, then we can circumvent the iterations
in Step 3 and part of Step 1 as follows: We substitute Step 3 by rules

notok :− ninS(h1), . . . , ninS(hl), inS(b1), . . . , inS(bm), ninS(bm+1), . . . ninS(bn). (2)

for each ruler in Π of form (1). These rules can be efficiently generated in parallel to
F (Π). Lines 8 to 13 of Step 1 then can also be dropped.

We can even refine this further. For any normal ruler ∈ Π with |H(r)| = 1 which
has a satisfied body, we can force the guess ofh: we replace (2) by

inS(h) :− inS(b1), . . . , inS(bm), ninS(bm+1), . . . ninS(bn). (3)

In this context, since constraints only serve to “discard” unwanted models but cannot
prove any literal, we can ignore them during input generationF (Π); rule (2) is suffi-
cient. Note that dropping input representationlit(n, l, c). for literals only occurring in
the negative body of constraints but nowhere else inΠ requires some care. Suchl can
be removed by simple preprocessing, though.

(OPT2)Optimize guess of orderWe only need to guess and check the orderφ for lit-
eralsL, L′ if they allow for cyclic dependency, i.e., they appear in the heads of rules
within the same strongly connected component of the program wrt.S.5 These depen-
dencies wrt.S are easily computed:

dep(L, L1):−lit(h, L, R), lit(p, L1, R), inS(L), inS(L1).
dep(L, L2):−lit(h, L, R), lit(p, L1, R), dep(L1, L2), inS(L).
cyclic:−dep(L, L1), dep(L1, L).

The guessing rules forφ (line 18 and 19) are then be replaced by:

phi(L, L1) v phi(L, L1):−dep(L, L1), dep(L1, L), L < L1, cyclic.
phi(L, L2):−phi(L, L1), phi(L1, L2), cyclic.

Moreover, we add the new atomcyclic also to the body of the rules wherephi appears
(lines 36,40) to checkphi only if Π hasanycyclic dependencies wrt.S.

4 Integrating Guess and co-NP Check Programs
A general method for solvingNP problems using answer set programming is given
by the so called “guess and check” paradigm: First a (possibly disjunctive) program
is used to guess a set of candidate solutions, and then rules and constraints are added
which eliminate unwanted solutions. DLPs allow for the formulation of such problems
in a very intuitive way (e.g. solutions of 3-colorability, deterministic planning, etc.)
if checking is easy (often polynomial), such as checking whether no adjacent nodes
have the same color, a course of deterministic actions reaches a certain goal, etc. For
instance, given a graph as a set of facts of the formnode(x). andedge(x, y). we can
write a simple DLP which guesses and checks all possible 3-colorings as follows:

5 Similarly, in [1] φ : Lit(Π) → {1, . . . , r} is only defined for a ranger bound by the longest
acyclic path in any strongly connected component of the program.

col(red, X) v col(green, X) v col(blue, X) :− node(X). } Guess
:−edge(X, Y), col(C, X), col(C, Y). } Check

However, encoding problems where the check is in co-NP but not known to be poly-
nomial (or inNP) is not always obvious (e.g., for conformant planning [4], or minimal
update answer sets [5]). A simple, common workaround is to write two programs:

(i) a normal LP or HDLPΠguess, which guesses some solution;
(ii) a HDLP (equivalently, normal LP)Πcheck which encodes the co-NP check,

and proceed as follows: First compute, one by one, the candidate solutionsS1, S2, . . .
as answer sets ofΠguess; then, pipe eachSi as input toΠcheck; finally, outputSi if
Πcheck ∪ Si has no answer set.

By the computational power of full disjunctive logic programs (ΣP
2 [6]), we know

that such problems can also be expressed by a single EDLP,Πsolve. In the following,
we show how our transformationtr resp.trOpt from above can be used to automatically
combineΠguess andΠcheck into a single program.

We assume that the setLit(Πguess) is a Splitting Set [13] ofΠguess∪Πcheck, i.e. no
head literal fromΠcheck occurs inΠguess. This can be easily achieved by introducing
new predicate names, e.g.,p′ for a predicatep, and adding a rulep′(t):−p(t) in case.
Each ruler in Πcheck is of the form

h1 v · · · v hl :− bc1, . . . , bcm, not bcm+1, . . . , not bcn

bg1, . . . , bgp, not bgp+1, . . . , not bgq.
(4)

where thebgi are the body literals defined inΠguess. We writeBguess(r) for bg1, . . . ,
bgp, not bgp+1, . . . ,not bgq. We now define a new check program.

Program Π′
check contains the following rules and constraints:

1. The factsF (Πcheck) in a conditional version: For eachr∈Πcheck of form (4),

lit(h, l, r) :− Bguess(r). atom(l, |l|). for eachl ∈ H(r);
lit(p, bci, r) :− Bguess(r). for eachi ∈ {1, . . . , m};
lit(n, bcj , r) :− Bguess(r). for eachj ∈ {m + 1, . . . , n}.

2. each rule inΠmeta (where for the optimized version, in (2) and (3)Bguess(r) is
added to the bodies);

3. finally, a constraint:− not notok. This will eliminate all answer setsS such that
Πcheck ∪ S has an answer set.

The union ofΠguess and Π ′
check then amounts to the desired integrated encoding

Πsolve, which is expressed by the following result.

Theorem 3. For Πguess andΠcheck, the answer setsS′ of Πsolve = Πguess ∪Π ′
check

correspond 1-1 with the answer setsS of Πguess s.t.Πcheck ∪ S has no answer set.

Note that integrating guess and check programsΠguess andΠcheck, respectively,
to succeed iffΠcheck ∪ S hassomeanswer set, is easy. After ensuring the Splitting Set
property (if needed), simply takeΠsolve = Πguess∪Πcheck; its answer sets correspond
on the predicates inΠguess to the desired solutions.

9



5 Applications
We now exemplify the use of our transformation for twoΣP

2 -complete problems, which
thus involve co-NP-complete solution checking: one is about Quantified Boolean for-
mulas (QBFs) with one quantifier alternation, which are well-studied in Answer Set
Programming, and the other about conformant planning [4, 22]. Further examples of
such problems can be found e.g. in [6, 5, 11] (and solved similarly). However, note that
our method is applicable toanychecks encoded by inconsistency of some HDLP pro-
gram; co-NP-hardness is not a prerequisite.

5.1 Quantified Boolean Formulas

Given a QBFF = ∃x1 · · · ∃xm∀y1 · · · ∀yn Φ, whereΦ = c1∨· · ·∨ck is a propositional
formula overx1, . . . , xm, y1, . . . , yn in disjunctive normal form, i.e. eachci = `i,1 ∧
· · ·∧`i,il

and|`i,j | ∈ {x1, . . . , xm, y1 . . . , yn}, compute the assignments to the variable
x1, . . . , xm which witness thatF evaluates to true.

Intuitively, this problem can be solved by “guessing and checking” as follows:

(QBF g) Guess a truth assignment for the variablesx1, . . . , xm.
(QBF c) Check whether this assignment satisfiesΦ for all assignments ofy1, . . . , yn.

Both parts can be encoded by very simple HDLPs:

QBF g : QBF c :

x1 v−x1. ... xm v−xm. y1 v−y1. ... yn v−yn.

:− `1,1, . . . , `1,1l . ... :− `k,1, . . . , `k,kl .

Obviously, for any answer setS of QBF g, representing an assignment tox1, . . . , xn,
the programQBF c∪S has no answer set thanks to the constraints, iff every assignment
for y1, . . . , yn satisfies formulaΦ then. By the method sketched, we can now automat-
ically generate a single programQBF solve integrating the guess and check programs
(cf. Footnote 3). Note that the customary (but tricky) saturation technique to solve this
problem (cf. [6, 11]) is fully transparent to the non-expert.

5.2 Conformant planning

Loosely speaking, planning is the problem to find a sequence of actionsP = α1, α2,. . . ,
αn, a plan, which takes a system from an initial states0 to a statesn in which a goal
(often, given by an atomg) holds, where a states is described by values of fluents, i.e.,
predicates which might change over time.Conformant planning[8] is concerned with
finding a planP which works under all contingencies that may arise from incomplete
information about the initial state and/or nondeterministic action effects, which is inΣP

2

under certain restrictions, cf. [4, 22]. Hence, the problem can be solved with a guess and
(co-NP) check strategy.

As an example, we consider a simplified version of the well-known “Bomb in the
Toilet” planning problem (cf. [4, 17]): We have been alarmed that a possibly armed
bomb is in a lavatory which has a toilet bowl. Possible actions are dunking the bomb
into the bowl and flushing the toilet. After just dunking, the bomb may be disarmed or
not; only flushing the toilet guarantees that it is really disarmed.

Using the following guess and check programsBombg andBombc , respectively,
we can compute a plan for having the bomb disarmed by two actions:

10



Bombg : Bombc :

% Timestamps: % Initial state:
time(0). time(1). armed(0) v -armed(0).
%Guess a plan: % Frame Axioms:
dunk(T) v -dunk(T):−time(T). armed(T1):−armed(T), time(T),
flush(T) v -flush(T):−time(T). not -armed(T1), T1 = T + 1.

% Forbid concurrent actions: dunked(T1):−dunked(T), T1 = T + 1.

:− flush(T), dunk(T). % Effect of dunking:
dunked(T1):−dunk(T), T1 = T + 1.
armed(T1) v -armed(T1):−dunk(T),

armed(T), T1 = T + 1.
% Effect of flushing:
-armed(T1):−flush(T), dunked(T), T1 = T + 1.

% Check goal in stage 2:
:− not armed(2).

Bombg guesses all candidate plansP = α1, α2, using time points for action execu-
tion, while Bombc checks whether any such planP is conformant for the goalg =
not armed(2). Here, absence ofarmed(t) is viewed as -armed(t), i.e. CWA is used,
which saves a negative frame axiom on -armed. The final constraint eliminates a plan
execution iff it reaches the goal; thus,Bombc has no answer set iff the planP is con-
formant. Answer setS = {time(0), time(1), dunk(0), flush(1)} of Bombg corre-
sponds to the (single) conformant planP= dunk, flush for goalnot armed(2).

By our general method,Bombg andBombc can be integrated automatically into a
single programBombplan = Bombg ∪ Bombc ′ (cf. Footnote 3). It has a single answer
set, corresponding to the single conformant planP= dunk, flush as desired.

Note that our rewriting method is more generally applicable than the encoding for
conformant planning proposed by Leoneet al.[12] who require that state transitions are
specified by a positive constraint-free LP. Our method can still safely be used in pres-
ence of negation and constraints, provided action execution always leads to a consistent
successor state (cf. [4, 22] for a discussion).

6 Experiments

We have conducted some experiments to get an idea about the performance of our
automatically integrated encodings (even though this was not the major concern). To
this end, we made comparisons to hand-written integrated encodings (in particular, for
QBF evaluation using an ad hoc encoding from [11]) and to interleaved guess and check
programs (in particular, for conformant planning on variants of the Bomb in the Toilet
problem, executed on theDLVK planning system [4]). The results are shown in Table 1.

We have considered some random QBF instances withn existential andn universal
variables, denoted as QBF-n. For conformant planning, we have compared the inte-
grated encodings for some “Bomb in the Toilet” problems described in [4], where the
names BTC(i), BTUC(i) are defined, against theDLVK planning system [4], which
implements conformant planning by interleaved calls to separate guess and check pro-
grams. The columns in Table 1 report times for encodings using the basic approach
(meta), optimizationsOPT1, OPT2 and both (OPT ), respectively, compared with the
ad hoc encoding for QBFs and interleaved conformant plan computation usingDLVK; a

11



adhoc [11] meta OPT1 OPT2 OPT

QBF-4 0.01s 0.15s 0.10s 0.08s 0.06s
QBF-6 0.01s 1.08s 0.36s 0.17s 0.08s
QBF-8 0.01s 10.42s 1.43s 0.46s 0.10s
QBF-10 0.01s 60.74s 2.65s 1.32s 0.10s
QBF-12 0.01s - - 5.59s 0.11s
QBF-16 0.08s - - - 0.50s
Average times for 10 randomly chosen instances per size.

DLVK [4] meta OPT1 OPT2 OPT

BTC(2) 0.01s 1.16s 0.80s 0.15s 0.08s
BTC(3) 0.11s 9.33s 9.25s 8.18s 4.95s
BTC(4) 4.68s 71.3s 67.8s 333s 256s
BTUC(2) 0.01s 6.38s 6.26s 0.22s 0.17s
BTUC(3) 1.78s - - 28.12s 13.0s
BTUC(4) 577s - - - 2322s

Table 1.Experimental results for QBF (left) and Conformant Planning (right) for Bomb in Toilet

dash marks exceeding a time limit of 300s (QBFs), resp. 4000s (conformant planning).
We usedDLV 6 as a platform since other disjunctive ASP engines, in particular GNT,7

were significantly slower on all tested instances. More details and experiments are given
in the extended paper (cf. Footnote 3).

Clearly, the performance of the automatic integrated encodings was expected to
stay behind the other methods. Interestingly, for the QBF problem, the performance of
our optimized translation stays within reach of the ad hoc encoding in [11] for small
instances. For the planning problems, the integrated encodings tested still stay behind
the interleaved computation ofDLVK.

The results obtained usingDLVshow that the “guess and saturate” strategy in our ap-
proach benefits a lot from optimizations, but it might depend on the structure ofΠguess

andΠcheck, as well as on the heuristics used byDLV, which modifications yield gains.
We strongly believe that there is room for further improvements both on the translation
and for the underlyingDLV engine. We emphasize that the strength of our approach
appears if an integrated ad hoc encoding is non-obvious. Then, by our method such
an encoding can be generated automatically from separate guess and check programs,
which are often easy to formalize, while a manual integrated encoding may be difficult
to find (as in the case of conformant planning or minimal update answer sets [5]).

7 Conclusion

We presented a method for rewriting a head-cycle free (extended) disjunctive logic pro-
gram (HDLP)Π into a stratified disjunctive logic program without constraintstr(Π),
such that their answer sets correspond and a designated answer set oftr(Π) indicates
inconsistency ofΠ. Moreover, we showed how to use this method for automatically
integrating a guess and separate check program for a co-NP property (expressed by
inconsistency of an HDLP), into an equivalent single (extended) disjunctive logic pro-
gram. This reconciles pragmatic problem solving with the natural “guess and check”
resp. “Generate/Define/Test” approach in Answer Set Programming [11, 14], in case a
single program expressing the problem is difficult to write. In particular, it relieves the
ASP user from using tricky saturation techniques, as customary e.g. for QBF evaluation.

We consider our work as an initial step for further research on automatic integra-
tion of guess and check encodings which exploits the full expressive power of DLPs:
Integrated encodings like those considered are infeasible in less expressive frameworks
such as propositional SAT solving or normal logic programming. However, while ad
hoc encodings for specific problems are available, general methods in this direction
were still missing.

6 http://www.dlvsystem.com
7 http://www.tcs.hut.fi/Software/gnt/

Several issues remain for further work. Our rewriting method currently applies to
propositional programs. Thus, before transformation, the program should be instanti-
ated. A more efficient extension of the method to non-ground programs is needed, as
well as further improvements to the current transformations. Experimental results sug-
gest that structural analysis of the guess and check programs might be valuable for this.

A further issue are alternative transformations, possibly tailored for certain classes
of programs. Ben-Eliyahu and Dechter’s work [1], on which we build, aimed at trans-
forming HDLPs to SAT problems. It might be interesting to investigate whether related
methods such as the one developed for ASSAT [15], which was recently generalized by
Lee and Lifschitz [10] to disjunctive programs, can be adapted for our approach.

